

Total Functions for Automated Reasoning

Building Terminating Theorem Provers

Alexander Gryzlov

(IMDEA Software Institute)



Lambda World 2025
October 24, Cádiz, Spain

Agenda

1. AI & automated reasoning
2. Total programming
3. Unification
4. SAT & DPLL
5. Iterative DPLL

As you can guess/deduce from the title, this is a talk about combining
two topics I've been working on for the past couple of years.

Agenda

1. AI & automated reasoning
2. Total programming
3. Unification
4. SAT & DPLL
5. Iterative DPLL

Part 1

The domain:
AI & Automated Reasoning

AI through time

AI is somewhat like "teenage music".

For every decade, it could mean something different:

- 1960s - The Beatles & Expert systems
- 1990s - Backstreet Boys & Spam filter/recommenders
- 2020s - Ariana Grande & Large Language Models



Let us go back to the basics!

Classic AI tasks

Formulated back in 1950s as "human-level activities performed by computer:"

- Machine translation and text comprehension
- Speech and pattern recognition
- Game playing (chess, checkers, Go, etc)
- *Theorem proving*

We'll focus on the last one:

Here, symbolic (rather than statistical) methods are typically used, and precise guarantees matter the most.


Theorem proving landscape

- SAT/SMT solvers (Z3/cvc5)
- Computer algebra systems (Mathematica, MAPLE)
- Automated provers (Vampire, E, iProver)
- Logic programming languages (Prolog, λ Prolog, Datalog)
- Proof assistants (HOL, Coq, Agda, Lean)

There's a struggle between the power of the system and automation.

More versatile systems converge toward general-purpose programming languages.

Automation vs power



Shoham, [2019] "Verification of Distributed Protocols Using Decidable Logic"


We'll use an interactive tool (Agda) to build and verify some simple automated ones.

Automated reasoning

The subfield of symbolic AI focused on theorem proving is referred to as "automated reasoning."

Typically involves:

- operating on syntax with variables
- synthesizing terms/proofs/refutations (*not always*)
- computing finite representations of functions (maps)



The algorithms we'll see use first-order syntax:

no binders (like λ) inside terms.

Variables and contexts

An important notion when reasoning about variables: *context*.

This is what logicians/type theorists write as capital Greek letters (Γ/Δ).

A finite set of all variables in the expression.

An overapproximation - can include extra variables not in the term!

We'll use a special (quotient) type Ctx for sets of variables (the order and multiplicity in it don't matter).

Has the usual set operators and predicates: \in , union, rem, minus

Part 2

The technique:

Total programming


The nature of functions

We're going to compute maps/functions, but can we be precise about them?

“ a function from a set X to a set Y assigns to each element of X exactly one element of Y ”

(Pure) functions in FP are close to the mathematical definition of a function.

But how do we guarantee the "each" and "exactly one" part?



Computational functions

Two problems beyond purity:

- not every X is assigned a Y
- a Y is never produced

```
head :: [a] -> a
head []     = error "oops"
head (x:_)
```

```
loop :: a -> a
loop x = loop x
```

Termination matters

Can be costly in critical systems, we typically expect each request/component to finish, even if the system is interactive:

“ 916 such CVE’s between 2000 and 2022

"Large-scale analysis of non-termination bugs in real-world OSS projects" (2022)

For reasoning algorithms, this means we always get an answer (though it may take a long time).

Total programming

We can only write programs which:

- cover all inputs
- terminate

The *first* part is relatively trivial (though often requires restructuring your program),


however the *second* involves recursion :(

Structural recursion

Programs which "consume" syntactically smaller pieces of input:

```
data Tree :  $\mathcal{U}$  where
  leaf : Tree
  node : Tree → Tree → Tree

depth : Tree →  $\mathbb{N}$ 
depth leaf          = 0
depth (node l r)    = 1 + max (depth l) (depth r)
```



Beyond structural

Here's a simple example that doesn't fit this pattern:
Euclid's GCD algorithm

```
{-# TERMINATING #-}  
gcd : N → N → N  
gcd n m =  
  if m == 0  
    then n  
  else gcd m (n % m)
```

```
gcd(105, 30) → 105%30 = 15  
gcd(30, 15) → 30%15 = 0  
gcd(15, 0) = 15
```

We know $n \% m < m$ but this is not structural!

Well-founded recursion


We have to introduce a "measure" that decreases on a type T according to a *well-founded* order.

If you come up with any sequence $\dots < t_x < t_y < t_z < \dots$, it cannot decrease forever.

A canonical type with such an order is $(\mathbb{N}, <)$.

Any sequence will eventually end with 0.

Called a linear order, but orders can be branching.



Well-founded language

Let's introduce a special type for this goal:

```
record □_ (A : T → U) (x : T) : U where
  field call : (x : T) → y < x → A y
  -----
fix : (A : T → U)
  → ( {t : T} → □ A t → A t)
  → ( {t : T} → A t)
-- a non-total fixpoint would be
-- fix : (A → A) → A
```

$\square A$ means "A can only be called with an argument smaller than its index".

If the order is well-founded, we can implement the fixed-point combinator!

Well-founded language - sugar

```
-- implicit
fix : (A : T → U)
  → ({t : T} → □ A t → A t)
  → ({t : T} → A t)
```

```
-- sugar!
fix : (A : T → U)
  → ∀[ □ A ⇒ A ]
  → ∀[ A ]
```

```
-- explicit
fix : (A : T → U)
  → ((t : T) → □ A t → A t)
  → ((t : T) → A t)
```

```
-- sugar!
fix : (A : T → U)
  → Π[ □ A ⇒ A ]
  → Π[ A ]
```

Sometimes we want to hide the decreasing argument (make it implicit), other times it's crucial to computation.

Well-founded GCD

Here's an example of computing a GCD function like this
(uses the explicit form):

```
gcd-ty :  $\mathbb{N} \rightarrow \mathcal{U}$ 
gcd-ty x = (y :  $\mathbb{N}$ )  $\rightarrow$  y < x  $\rightarrow \mathbb{N}$ 

gcd-loop :  $\Pi[\ \Box\ gcd-ty \Rightarrow gcd-ty\ ]$ 
gcd-loop x rec y y<x =
  cased y = 0 of
     $\lambda$  where
      (yes y=0)  $\rightarrow$  x
      (no y $\neq$ 0)  $\rightarrow$ 
        rec .call
    -- it is safe to do the recursive call
    -- remainder is smaller
      (%-r-< x y
       ( $\nexists$ ->-< $ contra  $\leq 0 \rightarrow = 0$  y $\neq$ 0))
```

To kick-start the computation, we just need to decide which argument goes first:

```
gcd< :  $\Pi[\ gcd-ty\ ]$ 
gcd< = fix gcd-ty gcd-loop

gcd :  $\mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N}$ 
gcd x y =
  caset x  $\geq <$  y of
     $\lambda$  where
      (LT x<y)  $\rightarrow$  gcd< y x x<y
      (EQ x=y)  $\rightarrow$  x
      (GT y<x)  $\rightarrow$  gcd< x y y<x
```

Part 3

The tutorial level:
Unification

Unification

Sometimes called "a Swiss army knife operator".

A family of (semi)algorithms for solving equations.

Match a pattern with gaps in it against data to fill the gaps.

Many applications:

- logic programming
- type inference
- 1st order logic provers
- program synthesis
- ...



Most General Unifier

First-order MGU: a classical form described in Pierce's TAPL.

Given a pair of terms (or generally a list of pairs),
find a substitution that makes all of them equal (or fail):

1	A	$\stackrel{?}{=} A$	$\{ \}$
2	A	$\stackrel{?}{=} B$	FAIL
3	A	$\stackrel{?}{=} x$	$\{ x \mapsto A \}$
4	A	$\stackrel{?}{=} B \otimes C$	FAIL
5	x	$\stackrel{?}{=} B \otimes C$	$\{ x \mapsto B \otimes C \}$
6	$x \otimes B$	$\stackrel{?}{=} A \otimes y$	$\{ x \mapsto A, y \mapsto B \}$
7	x	$\stackrel{?}{=} x \otimes x$	FAIL
8	$[x$	$\stackrel{?}{=} y$	
9	$, y]$	$\stackrel{?}{=} A]$	$\{ x \mapsto A, y \mapsto A \}$

Terms & constraints

- Terms are just binary trees with two kinds of leaves
- Variables are an abstract type with equality (think \mathbb{N} /String)
- Constraints are pairs of terms

We need both internal and external substitution:

```
1 data Term :  $\mathcal{U}$  where
2   ``_ : Var → Term
3   _⊗_ : Term → Term → Term
4   sy  : String → Term
5
6   -- x ⊗ B
7 example : Term
8 example = `` x ⊗ sy "B"
9
10 Constr :  $\mathcal{U}$ 
11 Constr = Term × Term
```

```
1 -- internal
2 sub1 : Var → Term → Term → Term
3 sub1 v t (`` x) =
4   if v == x then t else `` x
5 sub1 v t (p ⊗ q) =
6   sub1 v t p ⊗ sub1 v t q
7 sub1 v t (sy s) =
8   sy s
9
10 subs1 : Var → Term → List Constr → List Constr
11
12 -- external
13 Sub :  $\mathcal{U}$ 
14 Sub = Map Var Term
```

Unification code

```
1 unify : List Constr → Maybe Subst
2 unify [] = just emptyM
3 unify ((tl, tr) :: cs) =
4   if tl == tr
5     then unify cs
6     else unifyHead tl tr cs
7
8 unifyHead : Term → Term
9           → List Constr → Maybe Subst
10 unifyHead (`` v) tr cs =
11   if occurs v tr then nothing
12   else map (insertM v tr) $
13     unify (subs1 v tr cs)
14 unifyHead tl (`` v) cs =
15   ... -- symmetrical
16 unifyHead (lx ⊗ ly) (rx ⊗ ry) cs =
17   unify ((lx , rx) :: (ly , ry) :: cs) -- adds constraints!
18 unifyHead _ _ _ =
19   nothing
```

Why does it terminate?

- Can't just count constraints - they increase for the \otimes case
- Option 1: context size (count variables) - substitution *removes* variables; however, the context size **stays the same** for \otimes
- Option 2: count total term size in constraints - *decreases* for \otimes (one level gets dismantled), but can **grow** for var case

```
tm-size : Term → ℕ
tm-size (p ⊗ q) = 1 + tm-size p + tm-size q
tm-size _         = 1

tm-sizes : List Constr → ℕ
```

Solution: combine 1 and 2!

Lexicographic order

We can combine two (or generally N) well-founded orders:

$$(a, b) < (x, y) := a < x \text{ OR } (a = x \text{ AND } b < y)$$

One component always decreases!

For unification this means:

- Either context decreases (var case),
- Or it stays the same but the total term size does (\otimes case)

```
1 Input :  $\mathcal{U}$ 
2 Input = Ctx × List Constr
3
4 wf-tm : Ctx → Term →  $\mathcal{U}$ 
5 wf-tm c t = vars t ⊆ c
6
7 wf-input : Input →  $\mathcal{U}$ 
8 -- each term in the constraint
9 -- list is WF
```

```
unify-ty :  $\mathbb{N} \times \mathbb{N} \rightarrow \mathcal{U}$ 
unify-ty (x , y) =
  (inp : Input)
  → wf-input inp
  → x = size (inp .fst)
  → y = term-sizes (inp .snd)
  → Maybe Sub
```

Terminating case for \otimes

```
1 -- before
2 unifyHead (lx  $\otimes$  ly) (rx  $\otimes$  ry) cs =
3   unify ((lx , rx) :: (ly , ry) :: cs)
4
5 -- after
6 unify-head-loop rec (ctx , cs) wf (lx  $\otimes$  ly) (rx  $\otimes$  ry) wl wr ex ey =
7   rec .call prf-<
8     (ctx , ls') prf-wf
9     refl refl
10  where
11    cs' : List Constr
12    cs' = (lx , rx) :: (ly , ry) :: cs
13    prf-< : (size ctx , term-sizes cs') < (size ctx , term-sizes cs)
14    prf-< = ...
15    prf-wf : wf-input (ctx , cs')
16    prf-wf = ...
```

Part 4

The main quest:

SAT & DPLL

The SAT problem

Classical constraint satisfaction task:

Given a boolean formula with variables,
find an assignment of variables that makes it true (or fail).

Compared to unification:

- we restrict the range of variables (only True/False)
- but we add a semantical constraint (Boolean evaluation)

The SAT problem examples

```
1 -- tautologies (true for every assignment)
2
3 True
4 P ∧ Q ⇒ P ∨ Q
5 ((P ⇒ Q) ⇒ P) ⇒ P -- aka Peirce's law
6
7 -- satisfiable (there is an assignment)
8
9 P ∧ Q ⇒ Q ∧ R          -- P = Q = True, R = False
10                         -- simplifies to True ⇒ False
11
12 -- unsatisfiable
13
14 P ∧ ¬P
```

Exponential search

- The most widely used approach is backtracking search
- Naively we can try each variable and flip a previous choice when getting False
- However that's 2^n operations where $n = \#variables$
- Can we do better?

Generally, in the worst case, no! :(

Cook–Levin theorem (1971): SAT is NP-complete
(btw, this is the birth of NP-completeness concept).

But some *heuristics* can make less-than-worst cases tractable.

DPLL

Davis–Putnam–Logemann–Loveland algorithm, 1961

- Still fundamentally performs backtracking search
- But uses 3 heuristics to prune the search substantially
- These are *unit propagation*, *pure literal rule* and *literal selection*

CNF

DPLL and similar algorithms assume the input is in a *conjunctive normal form*: a big conjunction of disjunctions of possibly negated literals (*clauses*)

$$\begin{aligned} & (A \vee \neg B) \\ \wedge \quad & (\neg C \vee D \vee E) \\ \wedge \quad & \dots \end{aligned}$$

We can always transform into one thanks to boolean reasoning principles (i.e. DeMorgan rule: $\neg(P \vee Q) = \neg P \wedge \neg Q$)

True	$\sim \emptyset$
False	$\sim ()$
$P \vee Q$	$\sim (P \vee Q)$
$P \wedge Q$	$\sim (P) \wedge (Q)$
$P \wedge Q \Rightarrow Q \wedge R$	$\sim (\neg P \vee \neg Q \vee R)$

CNF encoding

Unlike in unification, we push the well-formedness constraint into the literals:

```
data Lit ( $\Gamma$  : Ctx) :  $\mathcal{U}$  where
  Pos : (v : Var)  $\rightarrow$  v  $\in$   $\Gamma \rightarrow$  Lit  $\Gamma$ 
  Neg : (v : Var)  $\rightarrow$  v  $\in$   $\Gamma \rightarrow$  Lit  $\Gamma$ 

var : Lit  $\Gamma \rightarrow$  Var
var (Pos v _) = v
var (Neg v _) = v

positive : Lit  $\Gamma \rightarrow$  Bool
positive (Pos _ _) = true
positive _ = false
```

```
Clause : Ctx  $\rightarrow$   $\mathcal{U}$ 
Clause  $\Gamma$  = List (Lit  $\Gamma$ )

CNF : Ctx  $\rightarrow$   $\mathcal{U}$ 
CNF  $\Gamma$  = List (Clause  $\Gamma$ )

literals : CNF  $\Gamma \rightarrow$  List (Lit  $\Gamma$ )
literals = nub  $\circ$  concat
```

Unit propagation aka 1-literal rule

Heuristic 1: If a clause consists of a single literal, it must be true, propagate it through the formula:

$$\begin{array}{l} (\text{A} \vee \neg\text{B}) \\ \wedge (\text{C}) \\ \wedge \dots \end{array}$$

```
1 unit-clause : CNF  $\Gamma$  → Maybe (Lit  $\Gamma$ )
2 unit-clause [] = nothing
3 unit-clause ([] :: c) = unit-clause c
4 unit-clause ((x :: []) :: c) = just x
5 unit-clause ((_ :: _ :: _) :: c) = unit-clause c
6
7 unit-propagate : (l : Lit  $\Gamma$ ) → CNF  $\Gamma$  → CNF (rem (var l)  $\Gamma$ )
8 unit-propagate l [] = []
9 unit-propagate l (f :: c) =
10  if has l f
11    then unit-propagate l c
12    else delete-var (var l) f :: unit-propagate l c
13
14 one-lit-rule : CNF  $\Gamma$  → Maybe ( $\Sigma$ [ l : Lit  $\Gamma$ ] (CNF (rem (var l)  $\Gamma$ )))
15 one-lit-rule cnf =
16  map ( $\lambda$  l → l , unit-propagate l cnf) $
17  unit-clause cnf
```

Pure literal rule

Heuristic 2 (aka affirmative-negative rule):

The idea is to delete every literal that occurs strictly positively or strictly negatively (purely).

```
pure-literal-rule :  
  (c : CNF  $\Gamma$ )  
  → ( $\Sigma$ [ purelits : List (Lit  $\Gamma$ ) ]  
    (let vs = map var purelits in  
      (vs  $\setminus$   $\Gamma$ )  $\times$  CNF (minus  $\Gamma$  vs)))  
     $\cup$  ( $\forall$  {l} → l ∈ literals c → negate l ∈ literals c)  
  ...
```

Literal selection

The previous two rules try to eliminate guessing as much as possible, but eventually, we're going to have to guess a value.

This is where backtracking still happens.

The splitting rule guarantees a result after the pure literal one.

```
1 posneg-count : CNF  $\Gamma \rightarrow \text{Lit}$   $\Gamma \rightarrow \mathbb{N}$ 
2 posneg-count cnf l =
3   let m = count (has           l) cnf
4     n = count (has $ negate l) cnf
5   in
6   m + n
7
8 splitting-rule : (c : CNF  $\Gamma$ )
9   → Any positive (literals c)
10  → Lit  $\Gamma$ 
```

Putting it all together

```
1 Answer = Map Var Bool
2
3 dpll-loop : (CNF Γ → Maybe Answer)
4           → CNF Γ → Maybe Answer
5 dpll-loop rec cnf =
6   if null? cnf then just emptyM          -- trivially true
7   else if has [] cnf then nothing        -- trivially false
8   else
9     maybe
10    (maybe
11      (let l = splitting-rule cls in
12        map (either (insertLit l)
13              (insertLit (negate l))) $
14              rec (unit-propagate l cnf)
15              <+> rec (unit-propagate (negate l) cnf))
16              (λ (ls , c) → map (insertLits ls) $ rec c)
17              (pure-literal-rule cnf))
18              (λ (l , c) → map (insertLit l) $ rec c)
19              (one-lit-rule cnf))
```

Why does this terminate?

Context always decreases:

- For unit propagation by 1
- For pure literal by some $n \geq 1$
- For recursive call by 1

This is actually simpler than unification!

We don't need the lexicographic pair, just a single number:

```
DPLL-ty : ℕ → ℰ
DPLL-ty x =
  {Γ : Ctx}
  → x = size Γ
  → CNF Γ → Maybe Answer
```

Part 5

The boss fight:
Iterative DPLL

Iterative DPLL

- Again, the core of the algorithm is backtracking search
- However, the backtracking information is kept on the system stack
 - no tail recursion
- Actual implementations work tail-recursively (iteratively)

```
...
  (map (either (insertLit l)
                (insertLit (negate l))) $  

    rec (unit-propagate l cls)  

    <+> rec (unit-propagate (negate l) cls))
...

```

Trail blazing

- We make the stack a first-class object, usually called a *trail*
- Need add a flag do determine if the literal is guessed, which then serves as a backtrack point

```
data Flag :  $\mathcal{U}$  where
  guessed deduced : Flag

Trail : Ctx →  $\mathcal{U}$ 
Trail  $\Gamma$  = List (Lit  $\Gamma$  × Flag)

trail-lits : Trail  $\Gamma$  → List (Lit  $\Gamma$ )
trail-lits = map fst

trail→answer : Trail  $\Gamma$  → Answer
trail→answer =
  foldr emp λ (l , _) → upd (var l) (positive l)
```

Heuristics redux

We get rid of the pure literal rule but perform unit propagation in batches:

```
1 unit-subpropagate-loop : (CNF  $\Gamma \rightarrow \text{Trail}$   $\Gamma \rightarrow \text{CNF} \Gamma \times \text{Trail} \Gamma$ )
2                                $\rightarrow \text{CNF} \Gamma \rightarrow \text{Trail} \Gamma \rightarrow \text{CNF} \Gamma \times \text{Trail} \Gamma$ 
3 unit-subpropagate-loop rec cnf tr =
4   let cnf' = map (filter (not  $\circ$  trail-has tr  $\circ$  negate)) cnf
5   newunits = literals (filter (is-fresh-unit-clause tr) cnf')
6   in
7   if null newunits
8   then (cnf' , tr)
9   else loop cnf' (map ( $\lambda l \rightarrow l$  , deduced) newunits ++ tr)
```

Iterative DPLL loop

Then we either run into an inconsistency and have to backtrack:

```
1 backtrack : Trail Γ → Maybe (Lit Γ × Trail Γ)
2 backtrack [] = nothing
3 backtrack ((_, deduced) :: ts) = backtrack ts
4 backtrack ((p, guessed) :: ts) = just (p, ts)
5
6 dpli-loop : CNF Γ
7     → (Trail Γ → Maybe Answer)
8     → Trail Γ → Maybe Answer
9 dpli-loop cnf rec tr =
10    let (cnf', tr') = unit-subpropagate cnf tr in
11    if has [] cnf' then -- reached False
12        maybe
13            nothing
14            (λ (p, trb) → rec ((negate p, deduced) :: trb))
15            (backtrack tr)
16    else
17        ...
```

Iterative DPLL loop

Or we have to make a choice:

```
1 dpli-loop : CNF  $\Gamma$ 
2           → (Trail  $\Gamma$  → Maybe Answer)
3           → Trail  $\Gamma$  → Maybe Answer
4 dpli-loop cnf rec tr =
5   let (cnf' , tr') = unit-subpropagate cnf tr in
6   if has [] cnf' then
7     ...
8   else                  -- need to guess
9     let ps = unassigned cls tr' in
10    if null ps
11      then just (trail→answer tr')
12      else rec ((splitting-rule' cls ps , guessed) :: tr)
```

Why does any of this terminate?

For unit propagation, the measure is the count of literals still unused in the trail:
(x2 because of polarity)

```
y = 2 * size Γ - length tr
```

However this, only works when the trail is *unique* (invariant #1)

Main loop termination

For the guess case, the unused trail literals also work:

```
rec ((splitting-rule' cls ps , guessed) :: tr)
```

We take the old trail and add a new literal onto it, exhausting unused ones.

But what about the backtracking case?

```
rec ((negate p , deduced) :: trb)
```

Here `trb` is a suffix of the old trail - it shrinks!


We had the same situation in unification: two kinds of recursive calls, one decreases a measure, the other one increases.

Looks like we need to use the lexicographic product again, but with what?

Main loop termination - idea

- Intuitively, it's the search space that decreases for backtracking
- But the search tree is generated on the fly
- We need to find a proxy


If we look closely at this tree, we notice that we never return to discarded branches.
So what sticks around is *rejected literals*.



Main loop termination - measure

- However, we can't lump them all together - the deduced literals get discarded after backtracking.
- We need to keep a vector of these rejected sets.
- The first part of the measure, then, is a vector of counts of still-available literals.

The full measure is then a lexicographic product of a vector (N-ary product) of non-rejected assignments corresponding to the guessed level and the number of unused variables in the trace!



(*whew*)

DPLI termination invariants

For all of this to work, we need the old trail uniqueness invariant and a new one:

1. Variable + polarity should not repeat
2. If a guessed variable is in the trail, its negation doesn't appear before it

The rejected vector/stack also needs an invariant:

If a variable is on level n , its negation appears in the trail after dropping the first n guessed variables.

```
Uniq (trail-lits tr) ×  
  (forall x → (x , guessed) ∈ tr  
   → negate x ∉ tail-of x (trail-lits tr))  
  
forall x (f : Fin (size Γ))  
  → x ∈ lookup rj f  
  → negate x ∈ (trail-lits $ drop-guessed tr (count-guessed tr - fin→N f))
```

Just need to prove that all of these are preserved :/

DPLI termination type

```
1 DPLI-ty : { $\Gamma$  : Ctx}  $\rightarrow$  Vec  $\mathbb{N}$  (size  $\Gamma$ )  $\times$   $\mathbb{N}$   $\rightarrow$   $\mathcal{U}$ 
2 DPLI-ty { $\Gamma$ } (x , y) =
3   (tr : Trail  $\Gamma$ )
4    $\rightarrow$  Trail-invariant tr
5    $\rightarrow$  (rj : Rejectstack  $\Gamma$ )
6    $\rightarrow$  Rejectstack-invariant rj tr
7    $\rightarrow$  x = map ( $\lambda$  q  $\rightarrow$  2  $\cdot$  size  $\Gamma$   $\dashv$  size q) rj
8    $\rightarrow$  y = 2  $\cdot$  size  $\Gamma$   $\dashv$  length tr
9    $\rightarrow$  Maybe Answer
```

Conclusion

Lessons learned


- Writing out termination proofs forces you to understand how the algorithm actually works, in the form of its invariants.
- Control flow tricks make termination harder :(
- Once you determine the invariants, you have more freedom to restructure your algorithm, make it more modular, experiment with different representations, and so on.

Lessons learned

- More generally, from a mathematical point of view, automated reasoning is about syntax, finite sets, maps, and intricate order relations.
- And order theory is just a cut-down version of category theory, but that is a story for another time...

References

- Automated Reasoning @ SEP
- Shi, Xie, Li, Zhang, Chen, Li, [2022] "Large-scale analysis of non-termination bugs in real-world OSS projects"
- Cook, Podelski, Rybalchenko, [2011] "Proving program termination"
- Hoder, Voronkov, [2009] "Comparing unification algorithms in first-order theorem proving"
- Vardi, [2015] "The SAT Revolution: Solving, Sampling, and Counting"
- Harrison, [2009] "Handbook of Practical Logic and Automated Reasoning"



Contacts & repo

- <https://www.linkedin.com/in/alexgryzlov/>
- <https://software.imdea.org/~aliaksandr.hryzlou/>
- <http://clayrat.github.io/>
- <https://twitter.com/clayrat/>

<https://github.com/clayrat/lw25-talk>




Alexander Gryzlov

Research Software Engineer at IMDEA
Software Institute



SCAN ME



Acknowledgments

Partially funded by the European Union (GA 101039196). Views and opinions expressed are however those of the author(s) only and do not necessarily reflect those of the European Union or the European Research Council. Neither the European Union nor the European Research Council can be held responsible for them.